
In order to better optimize and debug OpenCL kernels, sometimes it is very helpful to look at the underlying assembly. This article

shows you the tools available in the Intel® SDK for OpenCL™ Applications (https://software.intel.com/en-us/intel-opencl) that

allow you to view assembly generated by the offline compiler for individual kernels, highlight the regions of the assembly code that

correspond to OpenCL C code, as well as attempts at a high level explain different portions of the generated assembly. We also give

you a brief overview of the register region syntax and semantics, show different types of registers, and summarize available

assembly instructions and data types that these instructions can manipulate on. We hope to give you enough ammunition to get

started. In the upcoming articles we will cover assembly debugging as well as assembly profiling with Intel® VTune™ Amplifier

(https://software.intel.com/en-us/intel-vtune-amplifier-xe/).

Let us start with a simple kernel:

This is as simple as kernels get. We are going to build this kernel in a Code Builder Session Explorer. Go ahead and create a new

session by going to CODE-BUILDER/OpenCL Kernel Development/New Session, copying the kernel above to an empty program.cl

file and then building it. If you have a 5  generation Intel processor (Broadwell) or a 6  generation Intel processor (Skylake), you will

notice that one of the artifacts being generated is program_empty.gen file. Go ahead and double-click on it. What you will see is

1 kernel void empty() {
2 }

th th

something like this:



The assembly for the kernel is on the right: let me annotate it for you:

Not much, but it is a start.

Now, let’s complicate life a little. Copy the following into program.cl:

After rebuilding the file you will notice program_meaning_of_life.gen file. After double clicking on it you will see something more

complex:

What you can do now is to click on different parts of the kernel on the left, and see different parts of assembly being highlighted:

Here are instructions corresponding to the beginning of the kernel:

1 // Start of Thread
2 LABEL0
3 (W)      and      (1|M0)        r2.6<1>:ud    r0.5<0;1,0>:ud    0x1FF:ud         // id:
4
5 // End of thread
6 (W)      mov      (8|M0)        r127.0<1>:ud  r0.0<8;8,1>:ud   {Compacted}                 // id:
7 send     (8|M0)        null          r127              0x27      0x2000010 {EOT}  // id:

1 kernel void meaning_of_life(global uchar* out)
2 {
3 out[31] = 42;
4 }

The body of the kernel:

And the end of the kernel:

We are going to rearrange the assembly to make it a little bit more understandable:

01 // Start of Thread
02 LABEL0
03 (W)      and      (1|M0)        r2.6<1>:ud    r0.5<0;1,0>:ud    0x1FF:ud         // id:
04 // r3 and r4 contain the address of out variable (8 unsigned quadwords – uq)
05 // we are going to place them in r1 and r2
06 (W)      mov      (8|M0)        r1.0<1>:uq    r3.0<0;1,0>:uq                   // id:
07
08
09 // Move 42 (0x2A:ud – ud is unsigned dword) into 32 slots (our kernel is compiled SIMD32)
10 // We are going to use registers r7, r10, r13 and r16, each register fitting 8 values
11 mov      (8|M0)        r7.0<1>:ud    0x2A:ud          {Compacted}                 // id:
12 mov      (8|M8)        r10.0<1>:ud   0x2A:ud          {Compacted}                 // id:
13 mov      (8|M16)       r13.0<1>:ud   0x2A:ud                          // id:
14 mov      (8|M24)       r16.0<1>:ud   0x2A:ud                          // id:
15
16 // Add 31 (0x1F:ud) to eight quadwords in r1 and r2 and place the results in r3 and r4

Now, we are going to complicate life ever so slightly, by using get_global_id(0) instead of a fixed index to write things out

Note, that the addition of get_global_id(0) increases the size of our kernel by 9 assembly instructions. This mainly has to do with the

fact that we will need to calculate increasing addresses for each subsequent workitem in a thread (there 32 work items there):

17 // Essentially, we get &out[31]
18 (W)      add      (8|M0)        r3.0<1>:q     r1.0<0;1,0>:q     0x1F:ud          // id:
19
20 // Now we spread &out[31] into r5,r6, r8,r9, r11, r12, and r14, r15 – 32 values in all.
21 mov      (8|M0)        r5.0<1>:uq    r3.0<0;1,0>:uq                   // id:
22 mov      (8|M8)        r8.0<1>:uq    r3.0<0;1,0>:uq                   // id:1
23 mov      (8|M16)       r11.0<1>:uq   r3.0<0;1,0>:uq                   // id:1
24 mov      (8|M24)       r14.0<1>:uq   r3.0<0;1,0>:uq                   // id:1
25
26 // Write to values in r7 into addresses in r5, r6, etc.
27 send     (8|M0)        null          r5                0xC       0x60680FF                 
28 send     (8|M8)        null          r8                0xC       0x60680FF                 
29 send     (8|M16)       null          r11               0xC       0x60680FF                 
30 send     (8|M24)       null          r14               0xC       0x60680FF                 
31
32 // End of thread
33 (W)      mov      (8|M0)        r127.0<1>:ud  r0.0<8;8,1>:ud   {Compacted}                 // id:
34 send     (8|M0)        null          r127              0x27      0x2000010 {EOT}                 

1 kernel void meaning_of_life2(global uchar* out)
2 {
3 int i = get_global_id(0);
4 out[i] = 42;
5 }

01 // Start of Thread
02 LABEL0
03 (W)      and      (1|M0)        r7.6<1>:ud    r0.5<0;1,0>:ud    0x1FF:ud         // id:
04
05 // Move 42 (0x2A:ud – ud is unsigned dword) into 32 slots (our kernel is compiled SIMD32)
06 // We are going to use registers r17, r20, r23 and r26, each register fitting 8 values
07 mov      (8|M0)        r17.0<1>:ud   0x2A:ud          {Compacted}                 // id:
08 mov      (8|M8)        r20.0<1>:ud   0x2A:ud          {Compacted}                 // id:
09 mov      (8|M16)       r23.0<1>:ud   0x2A:ud                          // id:
10 mov      (8|M24)       r26.0<1>:ud   0x2A:ud                          // id:
11 // get_global_id(0) calculation, r0.1, r7.0 and r7.3 will contain the necessary starting values
12 (W)      mul      (1|M0)        r3.0<1>:ud    r0.1<0;1,0>:ud    r7.3<0;1,0>:ud   // id:
13 (W)      mul      (1|M0)        r5.0<1>:ud    r0.1<0;1,0>:ud    r7.3<0;1,0>:ud   // id:
14 (W)      add      (1|M0)        r3.0<1>:ud    r3.0<0;1,0>:ud    r7.0<0;1,0>:ud   {Compacted} // id:
15 (W)      add      (1|M0)        r5.0<1>:ud    r5.0<0;1,0>:ud    r7.0<0;1,0>:ud   {Compacted} // id:1
16 // r3 thru r6 will contain the get_global_id(0) offsets; r1 and r2 contain 32 increasing values
17 add      (16|M0)       r3.0<1>:ud    r3.0<0;1,0>:ud    r1.0<8;8,1>:uw   // id:1
18 add      (16|M16)      r5.0<1>:ud    r5.0<0;1,0>:ud    r2.0<8;8,1>:uw   // id:1
19 // r8 and r9 contain the address of out variable (8 unsigned quadwords – uq)
20 // we are going to place these addresses in r1 and r2
21 (W)      mov      (8|M0)        r1.0<1>:uq    r8.0<0;1,0>:uq                   // id:1
22
23 // Move the offsets in r3 thru r6 to r7, r8, r9, r10, r11, r12, r13, r14
24 mov      (8|M0)        r7.0<1>:q     r3.0<8;8,1>:d                    // id:1
25 mov      (8|M8)        r9.0<1>:q     r4.0<8;8,1>:d                    // id:1
26 mov      (8|M16)       r11.0<1>:q    r5.0<8;8,1>:d                    // id:1
27 mov      (8|M24)       r13.0<1>:q    r6.0<8;8,1>:d                    // id:1
28
29 // Add the offsets to address of out in r1 and place them in r15, r16, r18, r19, r21, r22, r24, r25
30 add      (8|M0)        r15.0<1>:q    r1.0<0;1,0>:q     r7.0<4;4,1>:q    // id:1
31 add      (8|M8)        r18.0<1>:q    r1.0<0;1,0>:q     r9.0<4;4,1>:q    // id:1
32 add      (8|M16)       r21.0<1>:q    r1.0<0;1,0>:q     r11.0<4;4,1>:q   // id:2
33 add      (8|M24)       r24.0<1>:q    r1.0<0;1,0>:q     r13.0<4;4,1>:q   // id:2
34
35 // write into addresses in r15, r16, values in r17, etc.
36 send     (8|M0)        null          r15               0xC       0x60680FF                 

Introduction to GEN Assembly | Intel® Developer Zone

4 of 14



And finally, let’s look at a kernel that does, reading, writing and some math:

It will be translated to the following (note, that I rearranged some assembly instructions for better understanding):

37 send     (8|M8)        null          r18               0xC       0x60680FF                 
38 send     (8|M16)       null          r21               0xC       0x60680FF                 
39 send     (8|M24)       null          r24               0xC       0x60680FF                 
40
41 // End of thread
42 (W)      mov      (8|M0)        r127.0<1>:ud  r0.0<8;8,1>:ud   {Compacted}                 // id:
43 send     (8|M0)        null          r127              0x27      0x2000010 {EOT}                 

1 kernel void modulate(global float* in, global float* out) {
2 int i = get_global_id(0);
3
4 float f = in[i];
5 float temp = 0.5f * f;
6 out[i] = temp;
7 }

01 // Start of Thread
02 LABEL0
03 (W)      and      (1|M0)        r7.6<1>:ud    r0.5<0;1,0>:ud    0x1FF:ud         // id:
04
05 // r3 and r4 will contain the address of out buffer
06 (W)      mov      (8|M0)        r3.0<1>:uq    r8.1<0;1,0>:uq                     // id:
07 // int i = get_global_id(0);
08 (W)      mul      (1|M0)        r5.0<1>:ud    r0.1<0;1,0>:ud    r7.3<0;1,0>:ud   // id:
09 (W)      mul      (1|M0)        r9.0<1>:ud    r0.1<0;1,0>:ud    r7.3<0;1,0>:ud   // id:
10 (W)      add      (1|M0)        r5.0<1>:ud    r5.0<0;1,0>:ud    r7.0<0;1,0>:ud   {Compacted} // id:
11 (W)      add      (1|M0)        r9.0<1>:ud    r9.0<0;1,0>:ud    r7.0<0;1,0>:ud   {Compacted} // id:
12 add      (16|M0)       r5.0<1>:ud    r5.0<0;1,0>:ud    r1.0<8;8,1>:uw   // id:
13 add      (16|M16)      r9.0<1>:ud    r9.0<0;1,0>:ud    r2.0<8;8,1>:uw   // id:
14
15 // r1 and r2 will contain the address of in buffer
16 (W)      mov      (8|M0)        r1.0<1>:uq    r8.0<0;1,0>:uq                   // id:1
17 // r11, r12, r13, r14, r15, r16, r17 and r18 will contain 32 qword offsets
18 mov      (8|M0)        r11.0<1>:q    r5.0<8;8,1>:d                    // id:1
19 mov      (8|M8)        r13.0<1>:q    r6.0<8;8,1>:d                    // id:1
20 mov      (8|M16)       r15.0<1>:q    r9.0<8;8,1>:d                    // id:1
21 mov      (8|M24)       r17.0<1>:q    r10.0<8;8,1>:d                   // id:1
22
23 //  float f = in[i];
24 shl      (8|M0)        r31.0<1>:uq   r11.0<4;4,1>:uq   0x2:ud           // id:1
25 shl      (8|M8)        r33.0<1>:uq   r13.0<4;4,1>:uq   0x2:ud           // id:1
26 shl      (8|M16)       r35.0<1>:uq   r15.0<4;4,1>:uq   0x2:ud           // id:1
27 shl      (8|M24)       r37.0<1>:uq   r17.0<4;4,1>:uq   0x2:ud           // id:1
28 add      (8|M0)        r19.0<1>:q    r1.0<0;1,0>:q     r31.0<4;4,1>:q   // id:1
29 add      (8|M8)        r21.0<1>:q    r1.0<0;1,0>:q     r33.0<4;4,1>:q   // id:2
30 add      (8|M16)       r23.0<1>:q    r1.0<0;1,0>:q     r35.0<4;4,1>:q   // id:2
31 add      (8|M24)       r25.0<1>:q    r1.0<0;1,0>:q     r37.0<4;4,1>:q   // id:2
32 // read in f values at addresses in r19, r20, r21, r22, r23, r24, r25, r26 into r27, r28, r29, r30
33 send     (8|M0)        r27           r19               0xC       0x4146EFF                 
34 send     (8|M8)        r28           r21               0xC       0x4146EFF                 
35 send     (8|M16)       r29           r23               0xC       0x4146EFF                 
36 send     (8|M24)       r30           r25               0xC       0x4146EFF                 
37
38 // float temp = 0.5f * f; - 0.5f is 0x3F000000:f
39 //     We multiply 16 values in r27, r28 by 0.5f and place them in r39, r40
40 //     We multiple 16 values in r29, r30 by 0.5f and place them in r47, r48
41 mul      (16|M0)       r39.0<1>:f    r27.0<8;8,1>:f    0x3F000000:f     // id:3
42 mul      (16|M16)      r47.0<1>:f    r29.0<8;8,1>:f    0x3F000000:f     // id:3
43
44 //     out[i] = temp;
45 add      (8|M0)        r41.0<1>:q    r3.0<0;1,0>:q     r31.0<4;4,1>:q   // id:2
46 add      (8|M8)        r44.0<1>:q    r3.0<0;1,0>:q     r33.0<4;4,1>:q   // id:2
47 add      (8|M16)       r49.0<1>:q    r3.0<0;1,0>:q     r35.0<4;4,1>:q   // id:2

Typically, all instructions have the following form:

[(pred)] opcode (exec-size|exec-offset) dst src0 [src1] [src2]

(pred) is the optional predicate. We are going to skip it for now.

opcode is the symbol of the instruction, like add or mov (we have a full table of opcodes below.

exec-size is the SIMD width of the instruction, which of our architecture could be 1, 2, 4, 8, or 16. In SIMD32 compilation, typically

two instructions of execution size 8 or 16 are grouped into one.

exec-offset is the part that's telling the EU, which part of the ARF registers to read or write from, e.g. (8|M24) consults the bits 24-31

of the execution mask. When emitting SIMD16 or SIMD32 code like the following:

         mov  (8|M0)   r11.0<1>:q   r5.0<8;8,1>:d   // id:1

         mov  (8|M8)   r13.0<1>:q   r6.0<8;8,1>:d   // id:1

         mov  (8|M16)  r15.0<1>:q   r9.0<8;8,1>:d   // id:1

         mov  (8|M24)  r17.0<1>:q   r10.0<8;8,1>:d  // id:1

the compiler has to emit four 8-wide operations due to a limitation of how many bytes can be accessed per operand in the GRF.

dst is a destination register

src0 is a source register

src1 is an optional source register. Note, that it could also be an immediate value, like 0x3F000000:f (0.5) or 0x2A:ud (42).

src2 is an optional source register.

Each thread has a dedicated space of 128 registers, r0 through r127. Each register is 256 bits or 32 bytes.

In the assembly code above, we only saw one of these special registers, the null register, which is typically used as a destination for

48 add      (8|M24)       r52.0<1>:q    r3.0<0;1,0>:q     r37.0<4;4,1>:q   // id:3
49
50 mov      (8|M0)        r43.0<1>:ud   r39.0<8;8,1>:ud  {Compacted}                 // id:3
51 mov      (8|M8)        r46.0<1>:ud   r40.0<8;8,1>:ud  {Compacted}                 // id:3
52 mov      (8|M16)       r51.0<1>:ud   r47.0<8;8,1>:ud                  // id:3
53 mov      (8|M24)       r54.0<1>:ud   r48.0<8;8,1>:ud                  // id:3
54
55 // write into addresses r41, r42 the values in r43, etc.
56 send     (8|M0)        null          r41               0xC       0x6066EFF                 
57 send     (8|M8)        null          r44               0xC       0x6066EFF                 
58 send     (8|M16)       null          r49               0xC       0x6066EFF                 
59 send     (8|M24)       null          r52               0xC       0x6066EFF                 
60
61 // End of thread
62 (W)      mov      (8|M0)        r127.0<1>:ud  r0.0<8;8,1>:ud   {Compacted}                 // id:
63 send     (8|M0)        null          r127              0x27      0x2000010 {EOT}                 

send instructions used for writing and indicating end of thread. Here is a full table of other architecture registers:

Since our registers are 32 bytes wide and are byte addressable, our assembly has a register region syntax, to be able to access

values stored in these registers.

Below, we have a series of diagrams explaining how register region syntax works.

Here we have a register region r4.1<16;8,2>:w. The w at the end of the region indicates that we are talking about word (or two

bytes) values. The full table of allowable integer and floating datatypes is below. The origin is at r4.1, which means that we are

starting with the second word of register r4. The vertical stride is 16, which means that we need to skip 16 elements to start the

second row. Width parameter is 8 and refers to the number of elements in a row; Horizontal stride of 2 means that we are taking

every second element. Note, that we refer here to the content of both r4 and r5. The picture below summarizes the result:

In this example, let’s consider a register region r5.0<1;8,2>:w. The region starts at a first element of r5. We have 8 elements in a

row, row containing every second element, so the first row is {0, 1, 2, 3, 4, 5, 6, 7}.  The second row starts at offset of 1 word, or at

r5.2 and so it contains {8, 9, 10, 11, 12, 13, 14, 15}. The picture below summarizes the result:

Consider the following assembly instruction

add(16|M0) r6.0<1>:w r1.7<16;8,1>:b r2.1<16;8,1>:b

The src0 starts at r1.7 and has 8 consecutive bytes in the first row, followed by the second row of 8 bytes, which starts at r1.23.

The src1 starts at r2.1 and has 8 consecutive bytes in the first row, followed by the second row of 8 bytes, which starts at r2.17.

The dst starts at r6.0, stores the values as words, and since the instruction Add(16) will operate on 16 values, stores 16 consecutive

words into r6.

Let’s consider the following assembly instruction:

add(16|M0) r6.0<1>:w r1.14<16;8,0>:b r2.17<16;8,1>:b

Src0 is r1.14<16;8,0>:b, which means the we have the first byte sized value at r1.14, 0 in the stride value means that we are going

to repeat the value for the width of the region, which is 8, and the region continues at r1.30, and we are going to repeat the value

stored there 8 times as well, so we are talking about the following value {1,1,1,1,1,1,1,1, 4, 4, 4, 4, 4, 4, 4, 4}.

Src1 is r2.17<16;8,1>:b, so we actually start with 8 bytes starting from r2.17 and end up with the second row of 8 bytes starting from

r3.1.

The letter after : in the register  region signifies the data type stored there. Here are two tables summarizing the available integer

and floating point types:

The following tables summarize available assembly instructions:



Volume 7 of Intel Graphics documentation is available here:

Volume 7: 3D-Media-GPGPU (https://01.org/sites/default/files/documentation/intel-gfx-prm-osrc-bdw-

vol07-3d_media_gpgpu_3.pdf)

Full set of Intel Graphics Documentation is available here:

https://01.org/linuxgraphics/documentation/hardware-specification-prms (https://01.org/linuxgraphics/documentation/hardware-

specification-prms)



Robert Ioffe is a Technical Consulting Engineer at Intel’s Software and Solutions Group. He is an expert in OpenCL programming

and OpenCL workload optimization on Intel Iris and Intel Iris Pro Graphics with deep knowledge of Intel Graphics Hardware. He was

heavily involved in Khronos standards work, focusing on prototyping the latest features and making sure they can run well on Intel

architecture. Most recently he has been working on prototyping Nested Parallelism (enqueue_kernel functions) feature of OpenCL

2.0 and wrote a number of samples that demonstrate Nested Parallelism functionality, including GPU-Quicksort for OpenCL 2.0. He

also recorded and released two Optimizing Simple OpenCL Kernels videos and is in the process of recording a third video on

Nested Parallelism.

You might also be interested in the following:

GPU-Quicksort in OpenCL 2.0: Nested Parallelism and Work-Group Scan Functions (https://software.intel.com/en-us/articles

/gpu-quicksort-in-opencl-20-using-nested-parallelism-and-work-group-scan-functions)

Sierpiński Carpet in OpenCL 2.0 (https://software.intel.com/en-us/articles/sierpinski-carpet-in-opencl-20)

Optimizing Simple OpenCL Kernels: Modulate Kernel Optimization (https://software.intel.com/en-us/videos/optimizing-simple-

opencl-kernels-modulate-kernel-optimization)

Optimizing Simple OpenCL Kernels: Sobel Kernel Optimization (https://software.intel.com/en-us/videos/optimizing-simple-

opencl-kernels-sobel-kernel-optimization)

